AD NUMBER
ADB003797
LIMITATION CHANGES
TO: **Approved for public release; distribution is unlimited.**
FROM: **Distribution authorized to U.S. Gov't. agencies only; Test and Evaluation; APR 1975. Other requests shall be referred to Army Ballistic Research Laboratories, Attn: AMXBR-SS, Aberdeen Proving Ground, MD 21005.**
AUTHORITY
USAMC ltr, 19 Dec 1986

BRL MR 24

BRL

AD

MEMORANDUM REPORT NO. 2470

ANALYSIS OF STEREO-PHOTOGRAPHIC PROJECTIONS

Henry C. Dubin

April 1975

Distribution limited to US Government agencies only; Test and Evaluation; APR 75. Other requests for this document must be referred to Director, USA Ballistic Research Laboratories, ATTN: AMXBR-SS, Aberdeen Proving Ground, Maryland 21005.

USA BALLISTIC RESEARCH LABORATORIES
ABERDEEN PROVING GROUND, MARYLAND

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER BRL MEMORANDUM REPORT NO. 2470	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) ANALYSIS OF STEREO-PHOTOGRAPHIC PROJECTIONS		5. TYPE OF REPORT & PERIOD COVERED Final
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Henry C. Dubin		8. CONTRACT OR GRANT NUMBER(s)
9. PERFORMING ORGANIZATION NAME AND ADDRESS USA Ballistic Research Laboratories Aberdeen Proving Ground, MD 21005		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 1T662618AH80
11. CONTROLLING OFFICE NAME AND ADDRESS U.S. Army Materiel Command 5001 Eisenhower Avenue Alexandria, VA 22333		12. REPORT DATE APRIL 1975
		13. NUMBER OF PAGES 26
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) UNCLASSIFIED
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE N/A

16. DISTRIBUTION STATEMENT (of this Report)

Distribution limited to US Government agencies only; Test and Evaluation; April 1975. Other requests for this document must be referred to Director, USA Ballistic Research Laboratories, ATTN: AMXBR-SS, Aberdeen Proving Ground, Maryland 21005.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Flash X-rays
Spark Shadowgraphs
Stereo-photographic Projections
Orthogonal Projections

20. ABSTRACT (Continue on reverse side if necessary and identify by block number) atp

FORTRAN and BASIC subroutines are presented for obtaining three-dimensional Cartesian coordinates from stereometric, photographic projections. A special case is orthogonal flash radiographs or shadowgraphs. The main feature of the subroutine is that it works for any source and film configuration.

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

TABLE OF CONTENTS

I. INTRODUCTION

One of the principle roles of the Ballistic Research Laboratories (BRL) is to generate experimental data. Often projectile velocities, orientations and behind the target space-time distributions of fragments are required. A common technique for obtaining such measurements is the use of orthogonal flash x-rays or spark shadowgraphs. The basic concept is illustrated in Figure 1. Two localized radiation sources, S_1 and S_2, project respective shadows, F_1 and F_2, of the projectile on the plane photographic plates oriented at right angles to each other. It is a straightforward geometry problem to use these source and image coordinates to obtain the projectile's space coordinates.[1*] Let X_p, Y_p and Z_p be the projectile coordinates in a Cartesian frame, O, such as indicated in Figure 1. Conceptually, one obtains Y_p and Z_p in terms of X_p from the S_2, F_2 coordinates, and similarly one obtains X_p and Z_p in terms of Y_p from the S_1, F_1 coordinates. Thus one has two equations in two unknowns, X_p and Y_p.

Quite often, however, spatial constraints on an experimental setup do not allow one to construct orthogonal film planes with the radiation sources far enough away to sufficiently reduce dispersion due to the finite source size. In such cases it is often convenient to construct film planes at other than right angles. Another example of a nonrectilinear setup can be found when the geometry of the apparatus suggests the use of curvilinear coordinates. In many small arms ranges safety requirements dictate that the flight of the projectile be confined to the interior of a hollow cylindrical chamber. In these cases, it would be convenient to mount the film on the inner cylinder wall with the radiation entering the chamber through a relatively small aperture.

It would be desirable to have a single simple means of converting a pair of independent, simultaneous projections of an object to its three-dimensional coordinates, regardless of the source and film configuration. This report addresses a practical solution to this problem. Although the basic approach to be employed is not original to this author,[2] it is felt that the universality and simplicity of the analysis developed make it a useful tool for experimentors needing spatial coordinates of projectiles in confined ranges. In addition to the presentation of the general analysis, a working computer code and straightforward applications to two sample problems are presented.

**Superscript numerals refer to references found on page 21.*



Figure 1. Schematic for Orthogonal Projections

II. THE GENERALIZED APPROACH

In the generalized approach no particular constraints will be placed on either the location of the radiation sources or the location of the film images as long as all locations can be given with respect to the same coordinate system. Figure 2 depicts the geometry for two independent projections of a projectile where S1 and S2 are the radiation sources and F1 and F2 are the corresponding film images. The Cartesian components of these locations are denoted by the subscripts i and j with the interpretation $i,j = x,y,z$. In this report vectors are denoted by underlining, and the convention employed in this analysis is to sum over repeated indices which appear in a single term. For example,

$$S1_i \, F2_i = S1_x F2_x + S1_y F2_y + S1_z F2_z ,$$

and

$$S2_j \, F2_j \, F1_i = (S2_x F2_x + S2_y F2_y + S2_z F2_z) \, F1_i .$$

For convenience we introduce the unit vector along the direction from $\underline{S1}$ to $\underline{F1}$ as $\underline{n1}$, and similarly the unit vector from $\underline{S2}$ to $\underline{F2}$ is denoted by $\underline{n2}$. In terms of Cartesian components, $\underline{n1}$ and $\underline{n2}$ are given by

$$n1_i = (\underline{F1}-\underline{S1})_i [(\underline{F1}-\underline{S1})_j (\underline{F1}-\underline{S1})_j]^{-1/2}$$

$$n2_i = (\underline{F2}-\underline{S2})_i [(\underline{F2}-\underline{S2})_j (\underline{F2}-\underline{S2})_j]^{-1/2} \quad .$$

The location of the projectile (i.e., a distinguishable point on the projectile) is the point where the lines along $\underline{n1}$ and $\underline{n2}$ intersect. One would find, however, that the slightest error in any of the measurements would most likely result in the lines not intersecting. Consequently, the location of the projectile will be obtained by finding the place where the two lines along $\underline{n1}$ and $\underline{n2}$ are closest to each other.

The precise procedure is as follows. Starting at the point $\underline{S1}$ let us move along the direction $\underline{n1}$ a distance D1, and similarly from $\underline{S2}$ we move along $\underline{n2}$ a distance D2. The desired location is given by the values of the parameters D1 and D2 which minimize the distance between the points on the lines $\underline{n1}$ and $\underline{n2}$ corresponding to the distances D1 and D2. The convention for defining the projectile location is to take the midpoint of the line connecting these two closest points on $\underline{n1}$ and $\underline{n2}$. This convention is illustrated in Figure 3.



Figure 2. The Geometry of the Generalized Approach



Figure 3. The **Convention** for Defining the Projectile Location

III. THE CALCULATION

The Cartesian components of the two closest points on $\underline{n1}$ and $\underline{n2}$ are given by :

$$D1_i = D1 \cdot n1_i + S1_i \tag{1}$$

and

$$D2_i = D2 \cdot n2_i + S2_i, \tag{2}$$

where the subscripted variables are Cartesian coordinates and the non-subscripted variables are invariant scalars. In order to find the distance of closest approach we shall minimize the distance between $\underline{D1}$ and $\underline{D2}$ with respect to the independent parameters D1 and D2. The distance to be minimized is denoted by M:

$$M = [(D1 \cdot n1_i + S1_i - D2 \cdot n2_i - S2_i)(D1 \cdot n1_i + S1_i - D2 \cdot n2_i - S2_i)]^{1/2} .$$

In order to minimize M we have to differentiate with respect to D1 and D2. In doing this, one finds that these derivatives of M are proportional to 1/M, and hence will be indeterminate if the lines do in fact intersect. To avoid testing for this indeterminacy it is convenient to minimize the square of the distance between $\underline{D1}$ and $\underline{D2}$. For completeness M^2 is given by:

$$M^2 = D1^2 + D2^2 + 2 \cdot D1\ (\underline{S1}-\underline{S2})_i\ n1_i + 2 \cdot D2\ (\underline{S2}-\underline{S1})_i\ n2_i$$
$$- 2 \cdot D1 \cdot D2 \cdot n1_i n2_i + (\underline{S1}-\underline{S2})_i (\underline{S1}-\underline{S2})_i\ .$$

The test to be used, therefore, is

$$\frac{\partial M^2}{\partial D1} + \frac{\partial M^2}{\partial D2} = 0.$$

Since D1 and D2 are independent, we have the two conditions

$$\frac{\partial M^2}{\partial D1} = 0$$

and

$$\frac{\partial M^2}{\partial D2} = 0.$$

Explicitly,

$$0 = \frac{\partial M^2}{\partial D1} = 2 \cdot D1 - 2 \cdot D2 \cdot n1_i n2_i - 2SL_i n1_i \tag{3}$$

and

$$0 = \frac{\partial M^2}{\partial D2} = 2 \cdot D2 - 2 \cdot D1 \cdot n1_i n2_i + 2SL_i n2_i \tag{4}$$

where $\quad \underline{SL} = \underline{S1} - \underline{S2}$.

Solving equations (3) and (4) for D1 and D2 gives

$$D2 = D1 \cdot n1_i n2_i + SL_i n2_i$$

in terms of D1, and

$$D1 = \frac{(SL_i n2_i)(n1_j n2_j) - SL_i n1_i}{1 - (n1_i n2_i)^2} .$$

It is noted that $n1_i n2_i$ is simply the cosine of the angle between the two rays $\underline{n1}$ and $\underline{n2}$. Therefore, by defining the sums

$$COS = n1_i n2_i$$

$$SL1 \quad SL_i n1_i$$

and

$$SL2 = SL_i n2_i ,$$

D1 and D2 are given by

$$D1 = \frac{(COS)(SL2) - SL1}{1 - (COS)^2} \tag{5}$$

and

$$D2 = (D1)(COS) + SL2 . \tag{6}$$

Since $\partial(D1)/\partial(COS) \to \infty$ as $COS \to 1$, and $\partial(D1)/\partial(COS)$ is a minimum for COS = 0, the determination of D1 and D2 will be least sensitive to measurement errors when $\underline{n1}$ and $\underline{n2}$ are orthogonal.

Substituting the values of D1 and D2 obtained from equations (5) and (6) into equations (1) and (2) gives $\underline{D1}$ and $\underline{D2}$. The remaining step in the calculation is to obtain the coordinates of the midpoint, P, of the line segment connecting $\underline{D1}$ and $\underline{D2}$. The coordinates of $\underline{P}$ are obtained by arithmetically averaging the corresponding coordinates of $\underline{D1}$ and $\underline{D2}$:

$$P_i = (1/2)(D1_i + D2_i) .$$

Using equations (1) and (2) we have

$$P_i = (1/2)(D1 \cdot n1_i + D2 \cdot n2_i + S1_i + S2_i)$$

which is the desired result.

IV. SAMPLE APPLICATIONS

As the analysis was developed we saw that the only input data were the coordinates of the radiation sources and the coordinates of the film images. Since the radiation sources are fixed in space, their locations are constant and need to be determined only once for a given experimental setup. The film image locations, however, must be determined from two-dimensional pictures which have been removed from the apparatus. This is easily accomplished by having a fiducial mark on the film and an analytic relation between distances in the film plane and the three-dimensional coordinates in the range setup. The fiducial marks will generally be made by opaque objects such as wires or metal tabs fixed to the aparatus at the exposed surface of the film.

The first and simplest example is the case of orthogonal film plates. For the particular choice of Cartesian axes and orthogonal fiducial lines shown in Figure 1, the translation from film images to spatial coordinates is straightforward. We shall consider Film 1 only in order to illustrate the procedure. In practice one measures the distances from the image to the fiducial lines. The x-coordinate is the perpendicular distance from the z-directed fiducial plus the distance from the z-directed fiducial to the z-axis. Similarly, the z-coordinate is the perpendicular distance from x-directed fiducial plus the distance between the x-directed fiducial and the x-axis. Lastly, for any point on Film 1 in Figure 1 the y-coordinate is the constant zero.

A simple variation on the above example is the case where physical constraints do not allow one to set up the film planes at right angles to each other. As illustrated in Figure 4 the coordinate system can be chosen such that one of the film planes lies in a convenient Cartesian plane. Consequently, the x-z-plane images can be located as above. To locate the image F, in the other plane we shall use the fiducial lines labeled 1 and 2 in Figure 4. The film reading measurements are the orthogonal components of F in the plane of the film, F_1 and F_2, as shown. The coordinates of F relative to the Cartesian axes are

$$F_x = (L_1 + F_1) \cos \theta,$$

$$F_y = (L_1 + F_1) \sin \theta \qquad (7)$$



Figure 4. Nonorthogonal Film Planes

and

$$F_z = (L_2 - F_2),$$

where the convention in reading the film has been chosen such that F_1 and F_2 are negative. For the conventions used, the Cartesian components of F which are generated by equations (7) are used in Equation (1) with no further adaptations required.

The second example application to be presented is the case of cylindrical symmetry. As before the only requirement is to relate the two-dimensional film measurements to three-dimensional coordinates in the range. The arbitrary choice for the Cartesian frame is shown in Figure 5(a).

In this example the film is mounted around the inner surface of a circular hollow cylinder, with inner radius R. Using the length and sign convention defined in Figure 5(b), we get for the spatial coordinates of the film image F,

$$\begin{aligned} F_x &= R(1 - \cos \phi) \\ F_y &= -R \sin \phi \\ F_z &= L_z + F_1, \end{aligned} \tag{8}$$

where

$$\phi = \frac{F_2}{R} \text{ radians.}$$

Obviously a second simultaneous source and film image pair are needed, and the same procedure can be used. Equations (8) are the only adaptation required in order to apply the generalized approach to the set-up depicted in Figure 5.

V. THE COMPUTER PROGRAM

The computer program consists of two subroutines. The first, SUBROUTINE TUBES(N), reads and stores the radiation source coordinates for each of N pairs of discharge tubes. Each pair of tubes which are discharged simultaneously is referred to as a channel. The sample programs listed in Figures 6 and 7 and the Appendix can accomodate up to twelve channels. In addition, SUBROUTINE TUBES generates the components of the relative vector between sources, SL, for each channel. The input for this subroutine is the set of tube coordinates and channel numbers formatted by FORTRAN Statement 101 in Figure 6. The output is a printed list of the source coordinates and an internal storage of the required vectors and coordinates in common block/T/.



Figure 5. Film Reading in Cylindrical Symmetry

```
      SUBROUTINE TUBES(N)
CCCCCCCCCCCCCCCCCCCC
C
C     THIS SUBROUTINE READS THE COORDINATES OF THE X-RAY SOURCES FOR N
C     CHANNELS OF ORTHOGONAL PAIRS AND STORES THEM IN (SEE FORMAT 101)
C
      COMMON/T/T(12,2,3),SL(12,3)
C
C     FOR T(I,J,K)
C         I IS CHANNEL NUMBER
C         FOR J
C           1=HORIZONTAL TUBES
C           2=VERTICAL TUBES
C         FOR K - 1=X, 2=Y, 3=Z
C*********(NOTE. BOTH THE ORIGIN AND UNITS MUST BE THE SAME AS THOSE
C         USED FOR THE FILM READING)*****SL IS DEFINED IN STATEMENT 1***
C
CCCCCCCCCCCCCCCCCCCC
      WRITE( 6,100)
      DO 1 K=1,N
      READ(5,101)NC,((T(NC,I,J),J=1,3),I=1,2)
      WRITE( 6,101)NC,((T(NC,I,J),J=1,3),I=1,2)
      DO 1 L=1,3
    1 SL(NC,L)=T(NC,1,L)-T(NC,2,L)
  100 FORMAT(1H1,12X,17H HORIZONTAL TUBES,25X,14HVERTICAL TUBES,/,8H CHA
     CNNEL,7X,1HX,9X,1HY,9X,1HZ,19X,1HX,9X,1HY,9X,1HZ,/)
  101 FORMAT(4X,I2,4X,3F10.4,10X,3F10.4)
      RETURN
      END
```

Figure 6. A FORTRAN Listing for Subroutine TUBES

```
      SUBROUTINE FINDX(NC)
CCCCCCCCCCCCCCCCCCCC
C
C     INPUT (SEE FORMAT 100)
C     NC=CHANNEL NUMBER
C     SPOT(NC,4)=TIME CHANNEL NC WAS DISCHARGED
C     F(I,J)=FILM IMAGE COORDINATES,
C            WHERE I=1 FOR HORIZONTAL TUBE
C                  I=2 FOR VERTICAL TUBE
C                  J=1 FOR X, 2 FOR Y, 3 FOR Z
C
CCCCCCCCCCCCCCCCCCCC
      COMMON/X/SPOT(12,4),XMISS(12)/T/T(12,2,3),SL(12,3)
      DIMENSION F(2,3),V(2,3),C(2)
      READ(5,100)NC,SPOT(NC,4),((F(I,J),J=1,3),I=1,2)
  100 FORMAT(4X,I2,3X,7F10.4)
C     CALCULATE THE UNIT VECTORS FROM TUBES TO FILM IMAGES
C     AT STATEMENT 2
      C(1)=0.
      C(2)=0.
      DO 1 K=1,2
      DO 1 I=1,3
      V(K,I)=F(K,I)-T(NC,K,I)
    1 C(K)=C(K)+V(K,I)*V(K,I)
      C(1)=SQRT(C(1))
      C(2)=SQRT(C(2))
      DO 2 K=1,2
      DO 2 I=1,3
    2 V(K,I)=V(K,I)/C(K)
C     CALCULATE THE DOT PRODUCTS OF THE UNIT VECTORS WITH EACH OTHER
C     AND WITH SL
      DOT=XMISS(NC)=0.
      SL1=SL2=0.
      DO 33 I=1,3
      DOT=DOT+V(1,I)*V(2,I)
      SL1=SL1+V(1,I)*SL(NC,I)
      SL2=SL2+V(2,I)*SL(NC,I)
   33 CONTINUE
      D1=(SL2*DOT-SL1)/(1.-DOT*DOT)
      D2=D1*DOT+SL2
C     CALCULATE THE MIDPOINT OF THE MISS DISTANCE VECTOR
C                                  =THE PROJECTILE COORDINATES
CCCCC OUTPUT GOES TO COMMON/X/SPOT(12,4),
CCCCC        WHERE SPOT(NC,4)=TIME
CCCCC              SPOT(NC,1)=PROJECTILE-X
CCCCC              SPOT(NC,2)=PROJECTILE-Y
CCCCC              SPOT(NC,3)=PROJECTILE-Z
      DO 8 I=1,3
      A1=T(NC,1,I)+D1*V(1,I)
      A2=T(NC,2,I)+D2*V(2,I)
      SPOT(NC,I)=(A1+A2)/2.
    8 XMISS(NC)=XMISS(NC)+(A1-A2)*(A1-A2)
      XMISS(NC)=SQRT(XMISS(NC))
      RETURN
      END
```

Figure 7. A FORTRAN Listing for Subroutine FINDX

The second subroutine, FINDX(NC), reads the film image coordinates for a channel and calculates the projectile coordinates at the time the channel is discharged. The inputs for this subroutine are common block /T/ and data cards for the film image coordinates which are formatted according to FORTRAN Statement 100 in Figure 7. Care must be taken to make sure that the first index of F(I,J) in FINDX and the second index of T(I,J,K) in TUBES correspond to the same source-image pair. It is noted that the names "horizontal" and "vertical" which are used in Figures 6 and 7 simply serve to identify the independent source-image pairs and do not presuppose any particular experimental setup. In addition, the user will probably want to include a subroutine to convert the film image measurements to three-dimensional coordinates as discussed in Section IV.

Subroutine FINDX performs the calculations derived in Section III and stores the projectile coordinates in the array SPOT. For many applications it would be convenient to add a third index to this array in order to identify a round number or a particular fragment. In cases where fragment identification on the two films may be ambiguous, as in reference 3, the miss distance M given in equation (3) can serve as a discriminator for various combinations of film images. For convenience M is saved in COMMON/X/XMISS(12).

In their present forms the calling sequence for the subroutines TUBES and FINDX is as follows. TUBES must be the first subroutine called and need only be called once for a given experimental setup; FINDX is called for each pair of film images to be analyzed. The calling program must declare the common block/X/SPOT(12,4),XMISS(12), in which the results are stored. Figure 8 is a sample calling routine for five channels with input and output. An appendix is included which lists the subroutines TUBES and FINDX in BASIC for a Wang 2200 computer.

VI. SUMMARY

In this report a simple and universal analysis for converting independent stereo-photographic projections of a projectile to its three-dimensional coordinates has been developed. The input requirements are the radiation source and film image coordinates. Additionally, sample computer programs for carrying out the calculations have been presented. These programs can be easily adapted to obtain projectile axes by locating two points, and similarly, planes can be obtained by locating three points.

```
      COMMON/X/SPOT(12,4),XMISS(12)
      READ(5,101)N
      CALL TUBES(N)
      WRITE( 6,100)
      DO 1 I=1,N
      CALL FINDX(NC)
      WRITE( 6,101)NC,(SPOT(NC,J),J=1,4),XMISS(NC)
    1 CONTINUE
  100 FORMAT(1H1,8H CHANNEL,7X,1HX,9X,1HY,9X,1HZ,6X,4HTIME,6X,4HMISS)
  101 FORMAT(4X,I2,4X,4F10.4,E11.2)
      END
```

(a)

```
5
1          0.     500.      0.          300.    200.      0.
2       -100.     500.    200.          300.    200.    100.
3        100.     500.    200.          300.    200.    200.
4       -200.     500.    400.          300.    200.    300.
5        200.     500.    400.          300.    200.    400.
1   98.       71.4     0.     35.7   -300.     80.0    59.9
2  174.       92.3     0.     46.2   -300.    145.4   100.3
3  248.      -66.6     0.    158.3   -300.    200.1   149.9
4  353.      118.2     0.    127.2   -300.    246.1   207.7
5  498.     -299.8     0.    250.0   -300.    285.9   271.3
```

(b)

CHANNEL	HORIZONTAL TUBES X	Y	Z	VERTICAL TUBES X	Y	Z
1	0.0000	500.0000	0.0000	300.0000	200.0000	0.0000
2	-100.0000	500.0000	200.0000	300.0000	200.0000	100.0000
3	100.0000	500.0000	200.0000	300.0000	200.0000	200.0000
4	-200.0000	500.0000	400.0000	300.0000	200.0000	300.0000
5	200.0000	500.0000	400.0000	300.0000	200.0000	400.0000

CHANNEL	X	Y	Z	TIME	MISS
1	49.97	149.99	24.97	98	2.97E-02
2	24.99	174.99	100.08	174	1.10E-01
3	0.05	200.04	174.96	248	2.92E-02
4	-24.99	224.98	249.98	353	5.38E-02
5	-49.81	250.07	324.99	498	5.89E-02

(c)

Figure 8. (a) Calling Program, (b) Input List, (c) Output

ACKNOWLEDGEMENT

The author would like to thank Mr. John Jameson of the U.S. Army Biomedical Laboratories, Biophysics Division for suggesting the generalized approach used in this analysis.

REFERENCES

1. C. Grabarek, L. Herr, "X-Ray Multi-Flash System for Measurement of Projectile Performance at the Target (U)," Ballistic Research Laboratories Technical Note No. 1634, September 1966. (AD #807619)

2. J. Jameson, U.S. Army Biomedical Laboratories, Biophysics Division, Edgewood Arsenal, personal communication.

3. A.L. Arbuckle, E.L. Herr, A.J. Ricchiazzi, "A Computerized Method of Obtaining Behind-the-Target Data From Orthogonal Flash Radiographs(U)," Ballistic Research Laboratories Memorandum Report No. 2264, January 1973. (AD #908362L)

APPENDIX

This appendix lists the subroutines TUBES and FINDX and a calling program for these subroutines in BASIC. This version of the program is operational on a Wang 2200 computer with the input instructions and output shown on the CRT screen. The pertinent changes in notation are that SPOT(I,J) and XMISS(I) are given by P(I,J) and M(I), respectively.

```
 1 REM---CALLING AND OUTPUT PROGRAM---
10 COM T(12,6),S(12,3),P(12,4),M(12):PRINT HEX (03)
20 INPUT "NO. OF CHANNELS",N
30 GOSUB '100(N)
40 FOR C=1TO N:GOSUB '31(C):NEXT C
50 PRINT HEX(03):PRINTUSING 60:PRINT
60 %CHANNEL      X          Y          Z          TIME        MISS
70 %   ##      -#####.## -####.## -####.##   -####.##    -#.##↑↑↑↑
80 FOR C=1TO N:PRINTUSING 70,C,P(C,1),P(,2),P(C,3),P(C,4),M(C):NEXT C
90 GOTO 330
```

```
100 REM---SUBROUTINE TUBES---FOR N CHANNELS
110 DEFFN'100(N):SELECT PRING 005:PRINT HEX(03)
120 FOR K=1TO N:PRINT "COORDINATES FOR CHANNEL";K:INPUT "X1,Y1,Z1,X2,Y2,Z2"
    T(K,2),T(K,3),T(K,4),T(K,5),T(K,6)
130 FOR L=1TO 3:S(K,L)=T(K,L)-T(K,L+3):NEXT L
140 NEXT K:RETURN
```

```
200 REM---SOUBROUTINE FINDX---FOR CHANNEL C
210 DEFFN'31(C):SELECT PRINT 005:PRINT HEX(03):DIM F(2,3),V(2,3),C(2)
220 PRINT "FOR CHANNEL";C:PRINT
230 INPUT "X,Y,Z FOR FILM IMAGE 1", F(1,1),F(1,2),F(1,3):INPUT"X,Y,Z
    FOR FILM IMAGE @",F(2,1),F(2,2),F(2,3):INPUT "TIME",P(C,4)
240 C(L),C(2(=0
250 FOR K=1TO 2:FOR I=1TO 3:V(K,I)=F(K,I)-T(C,I+3*(K-1)):C(K)=C(K)+V(K,I)
    ↑2:NEXT I:NEXT K
260 C(1)=SQR(C(1))"C(2)+SQR(C(2))
270 FOR K=1TO 2:FOR I=1TO 3:V(K,I)/C(K):NEXT I:NEXT K
280 D,S1,S2,M(C)=0
290 FOR I=1TO 3:D=D+V(1,I)*V(2,I):S1=S1+V(L,I)*S(C,I):S2=S2+V(2,I)*S(C,I):
    NEXT I
```

```
300 D1=(S2*D-S1)/(1-D!2):D2=D1*D+S2
310 FOR I=1TO 3:A1=T(C,I)+D1*V(1,I):A2=T(C,I+3)+D2*V(2,I):P(C,I)=(A1+A2)/2:
    = M(C)+(A1-A2)↑2:NEXT I
320 M(C)=SQR(M(C)):RETURN
330 REM
```

DISTRIBUTION LIST

No. of Copies	Organization
2	Commander Defense Documentation Center ATTN: DDC-TCA Cameron Station Alexandria, VA 22314
1	Director of Defense Research and Engineering (OSD) Washington, DC 20305
1	Director Defense Advanced Research Project Agency 1400 Wilson Boulevard Arlington, VA 22209
1	Director Weapons Systems Evaluation Group Washington, DC 20305
1	Chairman Defense Science Board Washington, DC 20301
2	Commander US Army Materiel Command ATTN: AMCDMA N. Klein J. Bender 5001 Eisenhower Avenue Alexandria, VA 22333
1	Commander US Army Materiel Command ATTN: AMCRD, BG H. A. Griffith 5001 Eisenhower Avenue Alexandria, VA 22333
1	Commander US Army Materiel Command ATTN: AMCRD-T 5001 Eisenhower Avenue Alexandria, VA 22333

No. of Copies	Organization
1	Commander US Army Aviation Systems Command ATTN: AMSAV-E 12th and Spruce Streets St. Louis, MO 63166
1	Director US Army Air Mobility Research & Development Laboratory Ames Research Center Moffett Field, CA 94035
1	Commander US Army Electronics Command ATTN: AMSEL-RD Fort Monmouth, NJ 07703
1	Commander US Army Missile Command ATTN: AMSMI-R Redstone Arsenal, AL 35809
1	Commander US Army Tank Automotive Command ATTN: AMSTA-RHFL Warren, MI 48090
2	Commander US Army Mobility Equipment Research & Development Center ATTN: Tech Docu Cen, Bldg. 315 AMSME-RZT Fort Belvoir, VA 22060
1	Commander US Army Armament Command Rock Island, IL 61202
2	Commander US Army Frankford Arsenal ATTN: SARFA-J7400 SARFA-J7600 Philadelphia, PA 19137

DISTRIBUTION LIST

No. of Copies	Organization
1	Commander US Army Picatinny Arsenal ATTN: SARPA-DW-6 Dover, NJ 07801
1	Commander US Army Harry Diamond Laboratories ATTN: AMXDO-TI 2800 Powder Mill Road Adelphi, MD 20783
1	Commander US Army Materials & Mechanics Research Center ATTN: AMXMR-ED Watertown, MA 02172
1	Commander US Army Natick Laboratories ATTN: AMXRE-CP Natick, MA 01762
1	Commander US Army Armor School ATTN: Armor Agency Fort Knox, KY 40121
1	Commander US Army Field Artillery School ATTN: Field Artillery Agency Fort Sill, OK 73405
1	Commander US Army Aviation School ATTN: Aviation Agency Fort Rucker, AL 36362
1	Commander US Army Infantry School ATTN: Infantry Agency Fort Benning, GA 31905

No. of Copies	Organization
1	The Surgeon General Department of the Army Washington, DC 20315
2	Commander US Naval Weapons Center ATTN: Code 753 Code 12 China Lake, CA 93555
1	Director US Naval Research Laboratories ATTN: Code 8434, Mr. Ferguson Washington, DC 20380
1	Commandant US Marine Corps Washington, DC 20380
1	Director MCDEC ATTN: C, Gnd Cbt Div Quantico, VA 22134
1	AFATL (DLRW) Eglin AFB Florida 32542

Aberdeen Proving Ground

Marine Corps Ln Ofc
Cmdr, USAEA
ATTN: SAREA-BL-B
Director, USAMSAA